Exam December E22
Suggested solutions

E22, 01006, DTU Compute
Rev. 08.12.22, shsp

Problem 1

```
> restart: with(LinearAlgebra): with(plots):
```

Given matrix with arbitrary constant $a \in \mathbb{R}$

```
> A:=<a,1,2|1,a,1|2,1,a|1,2,1>
```

$$A := \begin{bmatrix} a & 1 & 2 & 1 \\ 1 & a & 1 & 2 \\ 2 & 1 & a & 1 \end{bmatrix} \quad (1.1)$$

Given vectors with arbitrary constants $b, c \in \mathbb{R}$

```
> v1:=<7,1,-3>;
  v2:=<4,b,c>
```

$$v1 := \begin{bmatrix} 7 \\ 1 \\ -3 \end{bmatrix}$$

$$v2 := \begin{bmatrix} 4 \\ b \\ c \end{bmatrix} \quad (1.2)$$

1)
Given a value

```
> a:=0
```

$$a := 0 \quad (1.3)$$

Solving the matrix equation $A \cdot x = v_1$:

```
> LinearSolve(<A|v1>);
  simplify(%)
```

$$\left[\begin{array}{c} -\frac{a^2 \_t_4 - 7a^2 - 4a\_t_4 - 5a + 4\_t_4 + 8}{(a-2)\left(a^2+2a-2\right)} \\ -\frac{2a\_t_4 - a + 2\_t_4 + 2}{a^2+2a-2} \\ -\frac{a^2\_t_4 + 3a^2 - 4a\_t_4 + 15a + 4\_t_4 - 12}{a^3 - 6a + 4} \\ \_t_4 \end{array}\right]$$

$$\left[\begin{array}{c} -\_t_4 - 2 \\ \_t_4 + 1 \\ -\_t_4 + 3 \\ \_t_4 \end{array}\right] \quad \textbf{(1.4)}$$

General solution

$$x = \left[\begin{array}{c} x_1 \\ x_2 \\ x_3 \\ x_4 \end{array}\right] = \left[\begin{array}{c} -2 \\ 1 \\ 3 \\ 0 \end{array}\right] + t \left[\begin{array}{c} -1 \\ 1 \\ -1 \\ 1 \end{array}\right], t \in \mathbb{R}$$

2)

Given new a value

```
> a:='a':
  a:=2
```

$$a := 2 \quad \textbf{(1.5)}$$

Reducing the augmented matrix of the matrix equation $A \cdot x = v_2$:

```
> RowOperation(<simplify(A)|v2>,1,1);
  RowOperation(%,[3,1],-1);
  RowOperation(%,[2,1]);
  RowOperation(%,[2,1],-2);
  RowOperation(%,2,-1/3);
  RowOperation(%,[1,2],-2);
```

$$\begin{bmatrix} 2 & 1 & 2 & 1 & 4 \\ 1 & 2 & 1 & 2 & b \\ 2 & 1 & 2 & 1 & c \end{bmatrix}$$

$$\begin{bmatrix} 2 & 1 & 2 & 1 & 4 \\ 1 & 2 & 1 & 2 & b \\ 0 & 0 & 0 & 0 & c-4 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 2 & b \\ 2 & 1 & 2 & 1 & 4 \\ 0 & 0 & 0 & 0 & c-4 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 2 & b \\ 0 & -3 & 0 & -3 & 4-2b \\ 0 & 0 & 0 & 0 & c-4 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 1 & 2 & b \\ 0 & 1 & 0 & 1 & -\frac{4}{3}+\frac{2b}{3} \\ 0 & 0 & 0 & 0 & c-4 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 1 & 0 & -\frac{b}{3}+\frac{8}{3} \\ 0 & 1 & 0 & 1 & -\frac{4}{3}+\frac{2b}{3} \\ 0 & 0 & 0 & 0 & c-4 \end{bmatrix} \quad \textbf{(1.6)}$$

An inconsistent system (so no solutions) when $c \neq 4$. There were no exceptional cases during the reduction to take into account. So, the matrix equation has solutions only when $c = 4$ for all $b \in \mathbb{R}$.

Problem 2

```
> restart: with(LinearAlgebra): with(plots):
```

Given blue figure mapped to a red figure by a linear map $f : G2 \rightarrow G2$



Given vectors in e-coordinates

```
> ev1:=<3,4>;
  ev2:=<0,2>
```

$$ev1 := \begin{bmatrix} 3 \\ 4 \end{bmatrix}$$

$$ev2 := \begin{bmatrix} 0 \\ 2 \end{bmatrix} \quad \textbf{(2.1)}$$

1)
Checking for linear independence:

```
> Rank(<ev1|ev2>)
```

$$2 \quad \textbf{(2.2)}$$

Since the rank of the matrix created from the two vectors as columns equals the number of vectors, then they are linearly independent. They both lie within *G2*, and two are needed to span this space, so $v = (v_1, v_2)$ constitutes a basis for *G2*.

Change-of-basis matrix from v- to e-coordinates

```
> eMv:=<ev1|ev2>
```

$$eMv := \begin{bmatrix} 3 & 0 \\ 4 & 2 \end{bmatrix} \quad \textbf{(2.3)}$$

2)
Reading from the graph, the two vectors map to:
$_e f(v_1) = (-1, 7)$
$_e f(v_2) = (-2, 2)$

Mapping matrix wrt. v- and e-basis, respectively:

```
> eFv:=<-1,7|-2,2>
```

$$eFv := \begin{bmatrix} -1 & -2 \\ 7 & 2 \end{bmatrix} \quad (2.4)$$

Mapping matrix wrt. e-basis:

```
> eFe:=eFv.eMv^(-1)
```

$$eFe := \begin{bmatrix} 1 & -1 \\ 1 & 1 \end{bmatrix} \quad (2.5)$$

3)

Given vector w in e-coordinates

```
> ew:=<1,4>
```

$$ew := \begin{bmatrix} 1 \\ 4 \end{bmatrix} \quad (2.6)$$

Its image $f(w)$ in e-coordinates:

```
> efw:=eFe.ew
```

$$efw := \begin{bmatrix} -3 \\ 5 \end{bmatrix} \quad (2.7)$$

Their lengths

```
> L_w:=sqrt(ew.ew);
  L_fw:=sqrt(efw.efw)
```

$$L\_w := \sqrt{17}$$

$$L\_fw := \sqrt{34} \quad (2.8)$$

Cosine of angle between w and $f(w)$:

$$\cos(\theta) = \frac{w \cdot f(w)}{|w| \cdot |f(w)|} \Leftrightarrow$$

```
> cos_theta:= cos(theta)=ew.efw/(L_w*L_fw):
  simplify(cos_theta)
```

$$\cos(\theta) = \frac{\sqrt{2}}{2} \quad (2.9)$$

So, $\theta = \frac{\pi}{4}$ or $-\frac{\pi}{4}$, meaning that the angle between w and $f(w)$ is $\frac{\pi}{4}$.

Repeating the angle calculation for a general vector $q = (x, y)$

```
> eq:=<x,y>
```

$$eq := \begin{bmatrix} x \\ y \end{bmatrix} \quad (2.10)$$

Its image $f(q)$

```
> efq:=eFe.eq
```

$$efq := \begin{bmatrix} x-y \\ x+y \end{bmatrix} \quad (2.11)$$

Their lengths

```
> L_q:=sqrt(eq.eq) assuming real;
  L_fq:=sqrt(efq.efq) assuming real;
```

$$L\_q := \sqrt{x^2+y^2}$$

$$L\_fq := \sqrt{(x-y)^2+(x+y)^2} \quad (2.12)$$

Cosine of angle between q and $f(q)$:

$$\cos(\theta) = \frac{q \bullet f(q)}{|q| \cdot |f(q)|} \Leftrightarrow$$

```
> cos_theta:=cos(theta)=eq.efq/(L_q*L_fq) assuming real;
  simplify(%);
```

$$cos\_theta := \cos(\theta) = \frac{x(x-y)+y(x+y)}{\sqrt{x^2+y^2}\sqrt{(x-y)^2+(x+y)^2}}$$

$$\cos(\theta) = \frac{\sqrt{2}}{2} \quad (2.13)$$

We see that the angle between an arbitrary proper vector and its image again becomes $\frac{\pi}{4}$, so the map is angle conserving.

Problem 3

```
> restart: with(LinearAlgebra): with(plots):
```

Given characteristic polynomial of a 3×3 real matrix A

$$P(\lambda) = (\lambda+2) \cdot \left(\lambda - 1 + \frac{i}{2}\right) \cdot \left(\lambda - 1 - \frac{i}{2}\right), \lambda \in \mathbb{C}$$

1)
Eigenvalues via the rule of zero product

```
> lambda1:=-2;
  lambda2:=1-I/2;
  lambda3:=1+I/2
```

$$\lambda1 := -2$$

$$\lambda2 := 1 - \frac{I}{2}$$

$$\lambda3 := 1 + \frac{I}{2} \tag{3.1}$$

2)
It is given that to eigenspace E_{-2} belongs vector

```
> u1:=<1,0,0>
```

$$u1 := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} \tag{3.2}$$

and to eigenspace $E_{1-\frac{i}{2}}$ belongs vector

```
> u2:=<0,I,-1>
```

$$u2 := \begin{bmatrix} 0 \\ I \\ -1 \end{bmatrix} \tag{3.3}$$

These two vectors are thus eigenvectors of A. So, they map to themselves with their corresponding eigenvalue as the proportionality constant, $A \cdot u_1 = \lambda_1 u_1$ and $A \cdot u_2 = \lambda_2 u_2$:

```
> A.u__1=lambda1*u1;
  A.u__2=lambda2*u2
```

$$A \cdot u_1 = \begin{bmatrix} -2 \\ 0 \\ 0 \end{bmatrix}$$

(3.4)

$$A \cdot u_2 = \begin{bmatrix} 0 \\ \frac{1}{2} + I \\ -1 + \frac{I}{2} \end{bmatrix} \quad \textbf{(3.4)}$$

3)
Since the matrix is real, imaginary eigenvectors always come in complex conjugated pairs. We can thus state that to eigenspace $E_{1+\frac{i}{2}}$ belongs vector:

```
> u3:=conjugate(u2)
```

$$u3 := \begin{bmatrix} 0 \\ -I \\ -1 \end{bmatrix} \quad \textbf{(3.5)}$$

Since the sum of geometric multiplicities equals the number of rows (3) in the matrix, the matrix is diagonalizable with its eigenvalues defining the diagonal of a diagonal matrix Λ written with respect to an eigenbasis defined by the change-of-basis matrix U with corresponding eigenvectors as columns.

```
> Lambda:=DiagonalMatrix(<lambda1,lambda2,lambda3>);
  U:=<u1|u2|u3>
```

$$\Lambda := \begin{bmatrix} -2 & 0 & 0 \\ 0 & 1 - \frac{I}{2} & 0 \\ 0 & 0 & 1 + \frac{I}{2} \end{bmatrix}$$

$$U := \begin{bmatrix} 1 & 0 & 0 \\ 0 & I & -I \\ 0 & -1 & -1 \end{bmatrix} \quad \textbf{(3.6)}$$

Rewriting to determine a mapping matrix A with the above eigenproperties:
$U^{-1} A\, U = \Lambda \Leftrightarrow A = U\, \Lambda\, U^{-1}$

```
> A:=U.Lambda.U^(-1)
```

$$A := \begin{bmatrix} -2 & 0 & 0 \\ 0 & 1 & -\frac{1}{2} \\ 0 & \frac{1}{2} & 1 \end{bmatrix} \quad \textbf{(3.7)}$$

Problem 4

```
> restart: with(LinearAlgebra): with(plots):
```

Given matrix

```
> A:=<-3,4|0,-2>
```

$$A := \begin{bmatrix} -3 & 0 \\ 4 & -2 \end{bmatrix} \tag{4.1}$$

1)
Eigenvalues and -vectors

```
> Eigenvectors(A,output=list)
```

$$\left[\left[-2, 1, \left\{\begin{bmatrix} 0 \\ 1 \end{bmatrix}\right\}\right], \left[-3, 1, \left\{\begin{bmatrix} -\frac{1}{4} \\ 1 \end{bmatrix}\right\}\right]\right] \tag{4.2}$$

So, A has the eigenvalues -2 with a corresponding eigenvector $(0, 1)$ and -3 with a corresponding eigenvector $\left(-\frac{1}{4}, 1\right)$.

2)
Given inhomogeneous system of linear differential equations:

$$x_1'(t) + 3x_1(t) = -1 + 6t$$
$$x_2'(t) - 4x_1(t) + 2x_2(t) = -8t$$

```
> eq1:=diff(x1(t),t)+3*x1(t)=-1+6*t;
  eq2:=diff(x2(t),t)-4*x1(t)+2*x2(t)=-8*t
```

$$eq1 := \frac{d}{dt} x1(t) + 3\, x1(t) = -1 + 6\, t$$

$$eq2 := \frac{d}{dt} x2(t) - 4\, x1(t) + 2\, x2(t) = -8\, t \tag{4.3}$$

The system matrix is identical to matrix A from question **1)**, so with those eigenvalues and -vectors, the general solution to the corresponding homogeneous system is

$$x(t) = c_1 e^{-2t} \begin{bmatrix} 0 \\ 1 \end{bmatrix} + c_2 e^{-3t} \begin{bmatrix} -\frac{1}{4} \\ 1 \end{bmatrix}, \; t \in \mathbb{R} \text{ for any } c_1, c_2$$

3)
Guessing a particular solution to the inhomogeneous system

```
> x1p:=a*t+b;
  x2p:=c*t+d
```

$$x1p := a\,t + b$$

$$x2p := c\,t + d \tag{4.4}$$

First guess into first equation:

```
> diff(x1p,t)+3*x1p
```

$$3\,a\,t + a + 3\,b \tag{4.5}$$

As per the identity theorem of polynomials, coefficients of same-degree terms are equal, so:

```
> L1:=  3*a=6;
  L2:=  a+3*b=-1
```

$$L1 := 3\,a = 6$$

$$L2 := a + 3\,b = -1 \tag{4.6}$$

Second guess into second equation:

```
> diff(x2p,t)-4*x1p+2*x2p
```

$$-4\,a\,t + 2\,c\,t - 4\,b + c + 2\,d \tag{4.7}$$

Again via the identity theorem:

```
> L3:=  -4*a+2*c=-8;
  L4:=  -4*b+c+2*d=0
```

$$L3 := -4\,a + 2\,c = -8$$

$$L4 := -4\,b + c + 2\,d = 0 \tag{4.8}$$

Solving for the coefficients of the guess:

```
> solve({L1,L2,L3,L4})
```

$$\{a = 2, b = -1, c = 0, d = -2\} \tag{4.9}$$

4)

The guess in question **3)** worked and we have the particular solution:

$$x_p(t) = \begin{bmatrix} 2\,t - 1 \\ -2 \end{bmatrix}, t \in \mathbb{R}$$

Via the Structural Theorem, $L_{inhom} = x_p + L_{hom}$, the general inhomogeneous solution set is:

$$L_{inhom}: \quad x(t) = c_1 \mathrm{e}^{-2\,t} \begin{bmatrix} 0 \\ 1 \end{bmatrix} + c_2 \mathrm{e}^{-3\,t} \begin{bmatrix} -\frac{1}{4} \\ 1 \end{bmatrix} + \begin{bmatrix} 2\,t - 1 \\ -2 \end{bmatrix}, t \in \mathbb{R} \text{ for any } c_1, c_2$$